EE324 CONTROL SYSTEMS LAB

Problem Sheet 5

Yatish Vaman Patil | 190070076

Question 1: Root Locus

Part A)

In this part, we are given a closed-loop transfer function connected in a unity negative feedback loop.

$$\mathrm{Ts} = \frac{10}{s^3 + 4s^2 + 5s + 10}$$

Now, in general unity negative feedback, we get Ts as

$$\mathrm{Ts} = \frac{\mathrm{Gs}}{1 + \mathrm{Gs}}$$

$$\mathrm{Gs} = \frac{\mathrm{N}}{\mathrm{D}}$$

$$\mathrm{Ts} = \frac{\mathrm{N}}{\mathrm{N} + \mathrm{D}}$$

From comparing two equations, we get Gs as

$$\mathrm{Gs} = \frac{10}{s(s^2 + 4s + 5)}$$

Now we can get the root locus using Scilab



Part B)

$$Gs = \frac{s+1}{(s^2)(s+3.6)}$$



Part C)

$$Gs = \frac{s+0.4}{(s^2)(s+3.6)}$$



Part D)

$$Gs = \frac{s + p}{s(s + 1)(s + 2)}$$

In this part, we are varying-parameter 'p.' Following are the observations for different values of parameter 'p.'

- When p=0,1,2 :
  One of the three poles of the system cancels, and Gs becomes the second-order open-loop transfer function.
  This system always remains stable with any value of gain K as poles stay in LHP

- A considerable positive value of p:
  This system always remains stable as all poles and zeros are in LHP, so the poles of the closed-loop function will be in LHP for any gain K

- A negative value of p(p<0):
  This system will become unstable for $K > K_{marginal}$ a specific gain $K_{marginal} > 0$ because one zero of Gs lies in RHP

Following are Root Locus For Different values of 'p':



p=0



p=1



p=3



p=-3



p=-2



p=10

Scilab code for the same:

```
s=%s;

// A Part

Gs1=syslin('c',10,s*(s^2+4*s+5));
evans(Gs1,200);
show_window(1);

// B Part

Gs2=syslin('c',s+1,s^2*(s+3.6));
evans(Gs2,200);
show_window(2);

// C Part

Gs3=syslin('c',s+0.4,s^2*(s+3.6));
evans(Gs3,200);
show_window(3);

// D Part

Gs4=syslin('c',s+10,s*(s+1)*(s+2));
evans(Gs4,500);
show_window(4);
```

Question 2: Root Locus Analysis and Stability of System

Part A)

Same Breakaway and Breakin points

We can implement symmetric poles around the origin

$$Gs = \frac{s^4 + 1}{s^4 - 1}$$



Part B)

The number of branches at breakaway and break-in points are more than 4

$$Gs = \frac{s^7}{s^7 + 1}$$



Part C)

Branches of Root locus coincide with their asymptotes

$$Gs = \frac{10}{s^6 + 1}$$



Part D)

Breakaway and Break-in points are complex numbers

Step 1] Transfer function with no zero and real, symmetric poles around jw axis

$$Gs = \frac{144}{(s-3)(s+3)(s+4)(s-4)}$$

$$Gs = \frac{144}{(s^2-9)(s^2-16)}$$

Step 2] Replace s^2 with –s^2

$$Gs = \frac{144}{(-s^2-9)(-s^2-4)}$$

$$Gs = \frac{144}{(s^2+9)(s^2+16)}$$

Step 3] Replaces 's' with 's-k'. Here k=2

$$Gs = \frac{144}{((s-2)^2+9)((s-2)^2+16)}$$



Scilab code for the same:

```
s=%s;

// Part A
Gs1=syslin('c',s^4+1,s^4-1);
scf(0);
evans(Gs1,200);
xs2png(0,'Q2_A.png');

// Part B
Gs2=syslin('c',s^7/(s^7+1));
scf(1);
evans(Gs2,200000000000);
xs2png(1,'Q2_B.png');

// Part C
Gs3=syslin('c',10,s^6+1);
scf(2);
evans(Gs3,200);
xs2png(2,'Q2_C.png');

// Part D
Gs4=syslin('c',1,(((s-2)^2+9)*((s-2)^2+16)));
scf(3);
evans(Gs4,2000);
xs2png(3,'Q2_D.png');
```

Question 3: Designing Proportional Gain Controller:

$$Gs = \frac{1}{s(s^2 + 3s + 5)}$$

```
Minimum Rise time is: 0.555000
Minimum Rise time occurs for k=15.000000
Rise Time of System is 1.5s for k=3.600000
```

Scilab code for the same:

```
s=%s;

k=0.1:0.1:15;
time=0.0:0.015:60;
rise_times=zeros(1,length(k));

for  n=1:length(k)
    k_n=k(n);
    Gs=syslin('c',k_n/(s^3+3*s^2+5*s+k_n));
    Ps=csim('step',time,Gs);
    t_90=time(find(Ps>0.9))(1);
    t_10=time(find(Ps>0.1))(1);
    rise_times(1,n)= t_90-t_10 ;
end

Tr_min=min(rise_times(1,2:length(k)));
k_min=k(find(rise_times == Tr_min));

printf('Minimum Rise time is: %f',Tr_min);
printf('\nMinimum Rise time occurs for k=%f',k_min);

temp=find(rise_times>1.5);
k_req=0.1*length(temp);

printf('\nRise Time of System is 1.5s for k=%f',k_req);
```

Question 4: Proportional Controller for desired steady-state error:

Part A)

$$G = \frac{0.11(s + 0.6)}{6s^2 + 3.6127s + 0.0572}$$

At s=0

Steady State Error = $\frac{1}{1+kG}$ =0.01





Value of roportional gain Kp for 1 percent steady state error: Kp=85.800000

Scilab code for the same:

```
s=%s;
Gs=0.11*(s+0.6)/(6*s^2+3.6127*s+0.0572);
Gs=syslin('c',Gs);
evans(Gs,200);

time=0.1:0.1:500;
Ps=csim('step',time,Gs);
scf(2);
plot(time,Ps);
xlabel("time");
ylabel("Response");
xtitle("Step response of Open Loop Transfer Function");

gainK=0.1:0.1:100;
for i=1:length(gainK)
    Es=1/(1+gainK(i)*0.066/0.0572);
    if Es==0.01
        Kp=gainK(i);
    end
end

printf("Value of roportional gain Kp for 1 percent steady state error:
Kp=%f",Kp);
```

Part B)

We can use the RH table for determining the marginal stability of the closed-loop system

$$G = \frac{0.11(s + 0.6)}{6s^2 + 3.6127s + 0.0572}$$

Characteristic Polynomial of the system is

P = $6s^2 + (3.6127 + 0.11k)s + (0.0572 + 0.066k)$

s^2	6	0.0572+0.066k	0
s^1	3.6127+0.11k	0	0
s^0	0.0572+0.066k	0	0

3.6127+0.11k > 0 ➔ k > -32.84

0.0572+0.066k > 0 ➔ k > -0.867

Hence For Marginal Stabilty, Gain K = -0.867

Question 5:

Part A)

In this part, we have to consider two systems. The first one is the third-order system with two dominant and one pole on the extreme left side of the origin

$$\mathrm{Gs1} = \frac{1}{\left(\frac{s}{20}+1\right)(s^2+2s+5)}$$

$$\mathrm{Gs2} = \frac{1}{s^2+2s+5}$$



Third-Order System



Second-Order System

Scilab code for the same:

```
s=%s;

// 3 Pole System

Gs1=syslin('c',1,(s^2+2*s+5)*(s/20+1));
scf(1);
evans(Gs1,500);

// 2 Pole System

Gs2=syslin('c',1,(s^2+2*s+5));
scf(2);
evans(Gs2,500);
```

Part B)

It is evident from the root locus of both the systems that the second-order system will always remain stable for any value of gain K, but the third-order system will become unstable after reaching the specific maximum value of gain K. We can find this value using the RH table as follow

$$\mathrm{Gs1} = \frac{20}{(\mathrm{s}+20)(s^2+2s+5)}$$

Characteristic Polynomial becomes

$$\mathrm{Ps} = \mathrm{s}^3 + 22s^2 + 45s + 100 + 20k$$

s^3	1	45	0
s^2	22	100+20k	0
s^1	$\frac{890 - 20\mathrm{k}}{22}$	0	0
s^0	100+20k	0	0

890 – 20k > 0

k < 44.5

Hence, for K up to 44.5, the unit step response of both systems will be similar



Fig. Stable System



Fig. Marginally Stable System



Fig. Unstable System

Scilab code for the same:

```
s=%s;

Gs1=syslin('c',1,(s^2+2*s+5)*(s/20+1));
time=0:0.001:150;
```

```
Ps1=csim('step',time,44*Gs1/(1+44*Gs1));
Ps2=csim('step',time,44.5*Gs1/(1+44.5*Gs1));
Ps3=csim('step',time,45*Gs1/(1+45*Gs1));

plot(time,Ps1,'r');
legends(['K=44'],[5],opt="lr");
scf(2);
plot(time,Ps2,'g');
legends(['K=44.5'],[3],opt="lr");
scf(3);
plot(time,Ps3,'b');
legends(['K=45'],[2],opt="lr");
xlabel("Time");
ylabel("Response");
title("Step Response");
```